

# gnuplot y L<sup>A</sup>T<sub>E</sub>X: dos buenos amigos

Luis Peralta de Arriba

24 de junio de 2008

## Índice

<b>1. Introducción</b>	<b>3</b>
1.1. ¿Qué es <i>gnuplot</i> ?	3
1.2. ¿Qué es <i>T<sub>E</sub>X</i> ?	3
1.3. ¿Qué es <i>L<sup>A</sup>T<sub>E</sub>X</i> ?	3
1.4. ¿Por qué usar <i>gnuplot</i> y <i>L<sup>A</sup>T<sub>E</sub>X</i> ?	3
1.5. ¿Por qué se llevan bien <i>gnuplot</i> y <i>L<sup>A</sup>T<sub>E</sub>X</i> ?	4
1.6. ¿Qué se pretende con este documento?	4
<b>2. El terminal de salida <i>latex</i></b>	<b>5</b>
2.1. Las bondades del terminal	5
2.2. Selección del terminal de salida	5
2.3. Inclusión de un gráfico en un documento de <i>L<sup>A</sup>T<sub>E</sub>X</i>	5
2.4. Optimizaciones de los gráficos	6
2.5. Un ejemplo de gráfico	7
2.6. Tipos de líneas del terminal	9
2.7. Tipos de puntos del terminal	10
<b>3. El terminal de salida <i>epslatex</i></b>	<b>11</b>
3.1. Las bondades del terminal	11
3.2. Selección del terminal de salida	11
3.3. Inclusión de un gráfico en un documento de <i>L<sup>A</sup>T<sub>E</sub>X</i>	12
3.4. Optimizaciones de los gráficos	12
3.5. Un ejemplo de gráfico	13
3.6. Tipos de líneas del terminal	15
3.7. Tipos de puntos del terminal	16
<b>4. Referencias</b>	<b>17</b>
<b>5. Contacto</b>	<b>18</b>

## Derechos de autor y aviso de licencia

Copyright © 2008 Luis Peralta de Arriba.

*Permission is granted to copy, distribute and/or modify this document under the terms of the GNU Free Documentation License, Version 1.2 or any later version published by the Free Software Foundation; with no Invariant Sections, no Front-Cover Texts, and no Back-Cover Texts.*

Se concede permiso para copiar, distribuir y/o modificar este documento bajo los términos de la Licencia de Documentación Libre de GNU, Versión 1.2 o cualquier otra versión posterior publicada por la Free Software Foundation; sin Secciones Invariantes, ni Textos de Cubierta Delantera, ni Textos de Cubierta Trasera.

## 1. Introducción

### 1.1. ¿Qué es *gnuplot*?

*gnuplot* (<http://www.gnuplot.info/>) es un programa de línea de comandos para la representación de funciones, datos y diagramas.

Para la utilización de *gnuplot* se dispone también de interfaces gráficos, como por ejemplo *Qgfe*.

### 1.2. ¿Qué es *T<sub>E</sub>X*?

*T<sub>E</sub>X* es un sistema de tipografía desarrollado por Donald Ervin Knuth. Su principal virtud son las capacidades que ofrece para la creación de textos científicos, aunque dispone de muchas otras.

### 1.3. ¿Qué es *L<sup>A</sup>T<sub>E</sub>X*?

*L<sup>A</sup>T<sub>E</sub>X* es un conjunto de macros programados por Leslie Lamport que simplifican el uso de *T<sub>E</sub>X*.

### 1.4. ¿Por qué usar *gnuplot* y *L<sup>A</sup>T<sub>E</sub>X*?

Porque son dos buenos programas.

Porque son software libre, es decir, sus usuarios gozan de las siguientes libertades:

- La libertad de usar el programa con cualquier propósito (libertad 0).
- La libertad de estudiar cómo funciona el programa y adaptarlo a sus necesidades (libertad 1). El acceso al código fuente es una condición previa para esto.
- La libertad de distribuir copias, con lo que pueden ayudar a sus vecinos (libertad 2).
- La libertad de mejorar el programa y hacer públicas las mejoras a los demás, de modo que toda la comunidad se beneficia (libertad 3). El acceso al código fuente es un requisito previo para esto.

Porque están disponibles en Internet de manera gratuita.

### 1.5. ¿Por qué se llevan bien *gnuplot* y $\text{\LaTeX}$ ?

Porque *gnuplot* permite exportar sus gráficos en numerosos formatos a través de la selección del terminal de salida adecuado. En concreto, entre todos estos terminales, existen algunos que nos permiten seguir disfrutando del lenguaje de  $\text{\LaTeX}$  y de la tipografía que estamos utilizando nuestro documento de  $\text{\LaTeX}$ .

### 1.6. ¿Qué se pretende con este documento?

Explicar en castellano el funcionamiento de dos de los terminales de *gnuplot* diseñados para su utilización con  $\text{\LaTeX}$  mediante la recopilación de la información disponible en diferentes textos de la documentación de *gnuplot* y a partir de la experiencia del autor en el uso de los mismos.

Con este documento no se pretende enseñar el funcionamiento de  $\text{\LaTeX}$  ni de *gnuplot*. Si está interesado en ello, puede recurrir a los siguientes manuales:

- Una Descripción de  $\text{\LaTeX}2_{\epsilon}$
- *gnuplot*: herramienta para gráficos de funciones y datos

No obstante, para un uso avanzado de *gnuplot*, es conveniente consultar la documentación del programa o los ejemplos disponibles en su página web.

## 2. El terminal de salida *latex*

### 2.1. Las bondades del terminal de salida *latex*

El terminal de salida *latex* permite al usuario obtener un fichero en formato *TEX* listo para una inclusión directa en un documento *TEX* sin tener que hacer uso de ningún tipo de paquete.

### 2.2. Selección del terminal de salida *latex* en *gnuplot*

Para seleccionar el terminal de salida *latex* en *gnuplot*, habrá que ejecutar las siguientes instrucciones:

```
set terminal latex
set output 'nombre_de_archivo.tex'
```

Con ellas conseguiremos un archivo denominado `nombre_de_archivo.tex` escrito en formato *TEX*.

### 2.3. Inclusión de un gráfico generado con el terminal de salida *latex* en un documento de *TEX*

La inclusión del gráfico obtenido se puede hacer mediante la instrucción `input`:

```
\input{nombre_de_archivo.tex}
```

aunque lo más habitual es hacerla dentro de un entorno `figure`, tal y como muestra el siguiente código:

```
\begin{figure}
  \begin{center}
    \input{nombre_de_archivo.tex}
    \caption{Descripción}
  \end{center}
\end{figure}
```

## 2.4. Optimizaciones de los gráficos generados con el terminal de salida *latex*

### 2.4.1. El separador decimal

Si se está utilizando el terminal *latex*, puede resultarle interesante fijar el formato de la numeración entre dos símbolos de dólar:

```
set format '$%g$'
```

De esta manera conseguiremos que, dependiendo del idioma seleccionado en el documento de *LT<sub>E</sub>X*, el separador decimal sea un punto (como en inglés) o una coma (como en español).

### 2.4.2. Posición relativa de las etiquetas

A veces, si la etiqueta seleccionada para el eje Y tiene una longitud grande, ésta puede superponerse a la gráfica. Este problema puede arreglarse modificando la posición relativa de dicha etiqueta con respecto a la posición de referencia con el siguiente código:

```
set ylabel '[r]{Etiqueta Y}'
```

donde [r] hace referencia a una alineación a la derecha (podrían utilizarse también otras opciones [l] o [c] para alineaciones a la izquierda y centradas respectivamente, aunque no solucionarían el problema que estamos tratando).<sup>1</sup>

De manera análoga, y por si pudiera ser necesario en alguna situación, se podría modificar la alineación también para la etiqueta del eje X:

```
set xlabel '[t]{Etiqueta X}'
```

donde en lugar de [t] pueden emplearse otras opciones como [b] y [c].<sup>2</sup>

---

<sup>1</sup>Si sabe inglés, ya se habrá dado cuenta de que estas letras hacen referencia a las palabras *right*, *left* y *center*.

<sup>2</sup>Otra vez del inglés: *top*, *bottom* y *center*.

### 2.4.3. Rotación de la etiqueta del eje Y

Otra solución al problema tratado en la sección anterior es la rotación de la etiqueta del eje Y, resultando muchas veces la opción más elegante.

Para hacerlo, habrá que utilizar el paquete *graphicx* en el documento de *LaTeX*, es decir, habrá que incluir la siguiente instrucción en el preámbulo de dicho documento:

```
\usepackage{graphicx}
```

Así, estaremos en disposición de utilizar una orden como la siguiente en *gnuplot*:

```
set ylabel '\rotatebox{90}{Etiqueta Y}'
```

## 2.5. Un ejemplo de gráfico realizado con el terminal de salida *latex*

A continuación, se muestra el código de un ejemplo de gráfico realizado con el terminal de salida *latex*:

```
#!/usr/bin/gnuplot

# Configuración del terminal de salida
set terminal latex

# Selección del fichero de salida
set output 'nombre_de_archivo.tex'

# Selección del formato de la numeración
set format '$%g$'

# Desactivación de la leyenda del gráfico
set key off

# Etiquetas de los ejes
set xlabel '$x$'
set ylabel '\rotatebox{90}{$\sin(x)$}'

# Selección de los intervalos de representación
set xrange [0:2*pi]
set yrange [-1.1:1.1]
```

```
# Numeración de los ejes
set xtics ('$0$' 0, \
           '$\frac{\pi}{2}$' pi/2, \
           '$\pi$' pi, \
           '$\frac{3 \pi}{2}$' 3*pi/2, \
           '$2 \pi$' 2*pi)
set ytics ('$-1$' -1, \
           '$0$' 0, \
           '$1$' 1)

# Representación de la función seno
plot sin(x) linetype 3
```




Figura 1: Ejemplo de gráfico realizado con el terminal de salida *latex*



## 2.6. Tipos de líneas del terminal de salida *latex*




Figura 2: Tipos de líneas del terminal de salida *latex*

## 2.7. Tipos de puntos del terminal de salida *latex*




Figura 3: Tipos de puntos del terminal de salida *latex*.

Los tipos 1 ( $\diamond$ ) y 3 ( $\square$ ) requieren la utilización del paquete *latsym* en el documento de *ETEX*.

## 3. El terminal de salida *epslatex*

### 3.1. Las bondades del terminal de salida *epslatex*

Mediante el terminal de salida *epslatex* el usuario obtendrá dos ficheros, uno en formato *TEX* y otro en EPS, que le proporcionarán un resultado que permitirá preservar los textos en formato *TEX* y aprovechar las distintas posibilidades que el formato EPS tiene.

Las principales ventajas que este terminal de salida presenta frente al terminal *latex* son que permite hacer gráficos a una mayor resolución (lo que puede ser crítico para funciones con muchas oscilaciones o “ruido”) y la posibilidad de utilizar colores en los gráficos.

### 3.2. Selección del terminal de salida *epslatex* en *gnuplot*

Para seleccionar el terminal de salida *latex* en *gnuplot*, habrá que ejecutar las siguientes instrucciones:

```
set terminal epslatex
set output 'nombre_de_archivo.tex'
```

Con ellas conseguiremos un archivo denominado `nombre_de_archivo.tex` escrito en formato *TEX* y otro llamado `nombre_de_archivo.eps` en formato EPS.

No obstante, si, como es habitual hoy en día, pensamos procesar el documento de *TEX* con el programa `pdflatex` en lugar de `latex`, nos resultará útil ejecutar la siguiente llamada al sistema al final de nuestro script de *gnuplot* (o introducirla como último comando si estamos trabajando en modo interactivo):

```
system 'epstopdf nombre_de_archivo.eps'
```

Así conseguiremos un fichero con nombre `nombre_de_archivo.pdf` y en formato PDF, del que ya no nos tendremos que preocupar más, pues según procesemos el documento *TEX* con `pdflatex` o `latex`, estos programas harán uso del archivo que les conviene a cada uno de ellos (`latex` usará el fichero EPS, mientras que `pdflatex` el PDF).<sup>3</sup>

---

<sup>3</sup>Para ejecutar esta llamada al sistema necesitaremos tener instalado el programa *epstopdf* (como es obvio). Esto no supondrá ningún problema a los usuarios de GNU/Linux, ya que dicho programa es un habitual de todas sus distribuciones.

### 3.3. Inclusión de un gráfico generado con el terminal de salida *epslatex* en un documento de *TEX*

Para incluir los gráficos del terminal *epslatex* en un documento *TEX* se necesita utilizar el paquete *graphicx* en el documento de *TEX*, es decir, se tiene que utilizar la siguiente instrucción en el preámbulo de dicho documento:

```
\usepackage{graphicx}
```

Una vez introducida dicha instrucción, ya puede incluir el fichero en formato *TEX* con `input`:

```
\input{nombre_de_archivo.tex}
```

aunque habitualmente esta instrucción estará en un entorno `figure` como el siguiente:

```
\begin{figure}
  \begin{center}
    \input{nombre_de_archivo.tex}
    \caption{Descripción}
  \end{center}
\end{figure}
```

Nótese que no hace falta hacer referencia a los archivos en formato EPS y PDF, ya que el fichero `nombre_de_archivo.tex` es el que se encarga de incluir el adecuado en el documento.

### 3.4. Optimizaciones de los gráficos generados con el terminal de salida *epslatex*

#### 3.4.1. El separador decimal

Si se está utilizando el terminal *epslatex*, puede resultarle interesante fijar el formato de la numeración entre dos símbolos de dólar:

```
set format '$%g$'
```

De esta manera conseguiremos que, dependiendo del idioma seleccionado en el documento de *TEX*, el separador decimal sea un punto (como en inglés) o una coma (como en español).

### 3.4.2. Gráficos en color

Para conseguir que los gráficos sean exportados por *gnuplot* en color, es necesario utilizar la opción *color* del terminal; esto se consigue añadiendo la palabra *color* en la activación del terminal de salida, tal y como indica la siguiente línea:

```
set terminal epslatex color
```

### 3.4.3. Las opciones del terminal de salida *epslatex*

Además de la opción *color* del terminal, existen otras de menor relevancia y que son por tanto menos utilizadas. Si desea conocerlas, las encontrará documentadas en el manual de *gnuplot*.

## 3.5. Un ejemplo de gráfico realizado con el terminal de salida *epslatex*

A continuación, se muestra el código de un ejemplo de gráfico realizado con el terminal de salida *epslatex*:

```
#!/usr/bin/gnuplot

# Configuración del terminal de salida con la opción de color
set terminal epslatex color

# Selección del fichero de salida
set output 'nombre_de_archivo.tex'

# Desactivación del estilo predeterminado para las superficies
unset surface

# Selección del estilo pm3d para superficies en color
set pm3d

# Selección del formato de la numeración
set format '$%g$'

# Selección del intervalos de representación del eje Z
set zrange [-1:1]

# Numeración del eje Z
```

```
set ztics (-0.5, 0, 0.5, 1)

# Modificación de la cantidad de puntos en el muestreo
set isosamples 100, 100

# Representación de una función
splot sin(sqrt(x**2+y**2))/sqrt(x**2+y**2)

# Llamada al sistema para exportar el fichero EPS a PDF
system 'epstopdf nombre_de_archivo.eps'
```




Figura 4: Ejemplo de gráfico realizado con el terminal de salida *epslatex*

### 3.6. Tipos de líneas del terminal de salida *epslatex*




Figura 5: Tipos de líneas del terminal de salida *epslatex*

### 3.7. Tipos de puntos del terminal de salida *epslatex*




Figura 6: Tipos de puntos del terminal de salida *epslatex*.



## 4. Referencias

- $\text{\LaTeX}$  and the gnuplot plotting program
- Documentación oficial de gnuplot
- gnuplot: herramienta para gráficos de funciones y datos
- Una Descripción de  $\text{\LaTeX}2_{\epsilon}$

## 5. Contacto

Si tiene alguna duda sobre el contenido de este manual, ha detectado una errata o simplemente tiene alguna sugerencia sobre el mismo, puede mandarle al autor un correo electrónico a [luis\\_peralta@usal.es](mailto:luis_peralta@usal.es).